Adafruit IO Kontrol Motor Stepper

Deskripsi

Kontrol putaran motor stepper DC melalui aplikasi web io.adafruit.com. Sebagai controllernya NodeMCU ESP8266 yang dihubungkan dengan modul driver dan motor stepper.

Cara Kerja

NodeMCU ESP8266 memanfaatkan layanan MQTT server yang disediakan oleh Adafruit IO akan memproses command yang diterima kemudian diterjemahkan menjadi sinyal kontrol putaran motor stepper.



Adafruit IO adalah sistem yang disediakan oleh Adafruit (pembuat modul – modul sensor, perangkat elektronik, robot dll) untuk membuat sebuah proyek/aplikasi IoT (Internet of Things) . Adafruit IO ini boleh dipakai oleh siapa saja karena sifatnya free.

Motor stepper yang dipakai adalah tipe 28BYJ-48 dengan tegangan 5V dan driver motor ULN2003. Motor stepper ini merupakan motor unipolar dengan 5 jalur koneksi.

Sebagai driver-nya kita pakai board ULN2003 yang compatible dengan motor stepper 28BYJ-48 sehingga kita tinggal ‘colokkan’ koneksi kabelnya saja.

Kebutuhan Bahan

- NodeMCU V3
- Motor stepper 28BYJ-48
- Modul driver ULN2003
- Kabel konektor
- Kabel micro USB

Kebutuhan Software

- Arduino IDE
- Adafruit IO

Rangkaian/Skematik



Koneksi NodeMCU dengan Nodul driver ULN2003 :

NodeMCU	Modul Driver ULN2003
D5	IN1
D6	IN2

D7	IN3
D8	IN4
GND	GND



Membuat Antarmuka di Adafruit

Silakan masuk ke halaman https://io.adafruit.com/



Jika belum mempunyai akun silakan membuat akun terlebih dahulu. Siapkan sebuah akun email aktif.

Klik pada “**Get Started for free**” di bagian pojok kanan atas.



Isi data – data yang diperlukan kemudian klik “**CREATE ACCOUNT**” dan selanjutnya ikuti petunjuknya.

Masuk ke menu **IO**.





Klik pada tombol “**Actions**” kemudian pilih “**Create a New Dashboard**”



Selanjutnya akan muncul jendela untuk memberi nama, silakan beri nama pada dashboard yang akan dibuat, misalnya “**Kontrol Stepper**” dan klik “**Create**”.



Setelah klik tombol “**Create**” , akan muncul dashboard baru dengan nama “**Kontrol Stepper**”.



Klik “**Kontrol Stepper**” kemudian pilih tombol “**+**” untuk membuat blok baru.



Setelah tombol “**+**” di-klik akan muncul tampilan bermacam ‘widget’. Ada tombol (Toggle), tampilan Gauge, Slider, Text dll. Pilih **Momentary Button** .



Selanjutnya muncul tampilan untuk memilih ‘**Choose feed**’ . Ini semacam “Topik” dalam komunikasi MQTT. Jadi supaya dikenali oleh ‘subscriber’ (penerima perintah) maka ‘publisher’ harus menentukan ‘topik’ nya. Silakan tulis “**Stepper**” kemudian klik tombol “**Create**”.



Setelah tombol “**Create**” diklik dan selanjutnya pilih feed **Stepper**:



Kemudian klik “**Next Step**” sehingga muncul tampilan **Block settings** :



Isi pada kolom yang tersedia :

- Block Title : - (opsional)
- Button Text : PUTAR KANAN
- Press Value : 1
- Release Value : 0

Klik tombol “**Create block**”



Tambahkan komponen **Momentari Button** lagi.



Selanjutnya muncul tampilan untuk memilih ‘**Choose feed**’ . Ini semacam “Topik” dalam komunikasi MQTT. Jadi supaya dikenali oleh ‘subscriber’ (penerima perintah) maka ‘publisher’ harus menentukan ‘topik’ nya, pilih feed **Stepper**:



Kemudian klik “**Next Step**” sehingga muncul tampilan **Block settings** :



Isi pada kolom yang tersedia :

- Block Title : - (opsional)
- Button Text : PUTAR KIRI
- Press Value : 2
- Release Value : 0

Klik tombol "**Create block**". Selesai pembuatan antarmuka di Adafruit.IO.



Untuk pengaturan tata letak dan ukuran silakan edit dengan klik tombol Setting, kalau tidak diubah juga tidak mengapa.

Selanjutnya kita perlu kode (token) untuk mengisi variabel username dan active key di program Arduino yang akan dibuat. Klik bagian "**Adafruit IO Key**" di pojok kanan atas.



Perhatikan untuk **Username** dan **Active Key** ini penting, karena nanti dipakai ketika pemrograman dengan Arduino IDE. Catat/simpan kode tersebut.

Nah untuk pembuatan antarmuka pada Adafruit sudah selesai, sekarang kita siapkan program Arduino IDE-nya.

Program/Source Code

Program pada proyek ini memerlukan library :

- **ESP8266WiFi.h**
- **Adafruit_MQTT.h**
- **Adafruit_MQTT_Client.h**

Buka/jalankan Arduino IDE kemudian buat lembar kerja baru. Tulis kode program berikut.

```
/****************************************
 * Project Kontrol Motor Stepper dg Adafruit
 * Board  : NodeMCU ESP8266 V3
 * Input  : Adafruit
 * Output : Stepper
 * 99 Proyek IoT
 * www.ardutech.com
 ****************************************/

#include <ESP8266WiFi.h>
#include "Adafruit_MQTT.h"
#include "Adafruit_MQTT_Client.h"
//---GANTI SESUAI DENGAN JARINGAN WIFI
//---HOTSPOT ANDA
#define WLAN_SSID       "ArdutechWiFi"   // Nama Hotspot/WiFi
#define WLAN_PASS       "12345678"       //  Password
#define AIO_SERVER      "io.adafruit.com"
#define AIO_SERVERPORT  1883                   // use 8883 for SSL
//---GANTI SESUAI DENGAN USER NAME ADAFRUIT  ANDA
#define AIO_USERNAME    "ArdutechJogja"
//---GANTI SESUAI DENGAN IO KEY ADAFRUIT ANDA
#define AIO_KEY         "aio_mAXJ90DJnEIBVj0IDmG6BIMZqMAW"
WiFiClient client;
Adafruit_MQTT_Client    mqtt(&client,    AIO_SERVER,    AIO_SERVERPORT,
AIO_USERNAME, AIO_KEY);
Adafruit_MQTT_Subscribe    stepper    =    Adafruit_MQTT_Subscribe(&mqtt,
AIO_USERNAME "/feeds/Stepper");


uint8_t IN1 = D5;
```

```
uint8_t IN2 = D6;
uint8_t IN3 = D7;
uint8_t IN4 = D8;

const uint16_t _delay = 5;
uint32_t x=0;
//=======================================================
void MQTT_connect() {
  int8_t ret;
  if (mqtt.connected()) {
    return;
  }

  Serial.print("Connecting to MQTT... ");

  uint8_t retries = 3;
  while ((ret = mqtt.connect()) != 0) { // connect will return 0 for connected
       Serial.println(mqtt.connectErrorString(ret));
       Serial.println("Retrying MQTT connection in 5 seconds...");
       mqtt.disconnect();
       delay(5000);  // wait 5 seconds
       retries--;
       if (retries == 0) {
         while (1);
       }
  }
  Serial.println("MQTT Connected!");
}
//====================
void sequence(bool a, bool b, bool c, bool d){  /* four step sequence to stepper motor */
```

```
  digitalWrite(IN1, a);
  digitalWrite(IN2, b);
  digitalWrite(IN3, c);
  digitalWrite(IN4, d);
  delay(_delay);
}
void CW(){
    sequence(HIGH, LOW, LOW, LOW);
    sequence(HIGH, HIGH, LOW, LOW);
    sequence(LOW, HIGH, LOW, LOW);
    sequence(LOW, HIGH, HIGH, LOW);
    sequence(LOW, LOW, HIGH, LOW);
    sequence(LOW, LOW, HIGH, HIGH);
    sequence(LOW, LOW, LOW, HIGH);
    sequence(HIGH, LOW, LOW, HIGH);
}
void CCW(){
    sequence(LOW, LOW, LOW, HIGH);
    sequence(LOW, LOW, HIGH, HIGH);
    sequence(LOW, LOW, HIGH, LOW);
    sequence(LOW, HIGH, HIGH, LOW);
    sequence(LOW, HIGH, LOW, LOW);
    sequence(HIGH, HIGH, LOW, LOW);
    sequence(HIGH, LOW, LOW, LOW);
    sequence(HIGH, LOW, LOW, HIGH);
}
//==================================
void setup()
{
  pinMode(IN1, OUTPUT); /* set four wires as output */
  pinMode(IN2, OUTPUT);
  pinMode(IN3, OUTPUT);
```

```
  pinMode(IN4, OUTPUT);
  Serial.begin(9600);
   delay(10);
  Serial.println(); Serial.println();
  Serial.print("Connecting to... ");
  Serial.println(WLAN_SSID);

  WiFi.begin(WLAN_SSID, WLAN_PASS);
  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  Serial.println();

  Serial.println("WiFi connected");
  Serial.println("IP address: ");
  Serial.println(WiFi.localIP());
  mqtt.subscribe(&stepper);
}
//========================================
void loop()
{
  MQTT_connect();
  Adafruit_MQTT_Subscribe *subscription;
  while ((subscription = mqtt.readSubscription(5000))) {
    if (subscription == &stepper) {
      Serial.print(F("Got: "));
      Serial.println((char *)stepper.lastread);
      if(strcmp((char*)stepper.lastread,"1")==0){
        for( int i=0;i<200;i++)
        CW();
```

```
        }
      else if(strcmp((char*)stepper.lastread,"2")==0){
        for( int i=0;i<200;i++)
        CCW();
        }
      }
    }
}
```

PERHATIKAN !

Ganti/sesuaikan variabel berikut :

- Nama jaringan WiFi/hotspot : **WLAN_SSID**
- Password jaringan WiFi/hotspot : **WLAN_PASS**
- Username Adafruit IO : **AIO_USERNAME**
- Active Key Adafruit IO : **AIO_KEY**

Simpan (**Save**) kemudian **Upload**. Pastikan tidak ada error, jika masih ada silakan cek penulisan dll kemudian perbaiki. (Program ini sudah diuji langsung dan sudah berjalan tanpa ada error)

Jalannya Alat

Siapkan jaringan WiFi/hotspot atau tethering dari HP dan aktifkan. Buka Serial Monitor, dari menu **Tools → Serial Monitor** kemudian seting baudrate pada nilai 9600.



Jika belum muncul tekan tombol reset (RST) pada board NodeMCU sehingga muncul tampilan sebagai berikut :



Setelah berhasil terhubung dengan jaringan WiFi sekarang kita kembali ke tampilan di Adafruit. Tekan tombol tombol “PUTAR KIRI” maka motor stepper akan berputar ke kiri.





Tekan tombol “PUTAR KANAN” maka motor stepper akan berputar ke kanan.

Selamat berkarya , semoga lancar dan bermanfaat.

Ardutech – *“Sahabat Inovasi Anda”*